

NAG Fortran Library Routine Document

F07GEF (SPPTRS/DPPTRS)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F07GEF (SPPTRS/DPPTRS) solves a real symmetric positive-definite system of linear equations with multiple right-hand sides, $AX = B$, where A has been factorized by F07GDF (SPPTRF/DPPTRF), using packed storage.

2 Specification

```
SUBROUTINE F07GEF(UPLO, N, NRHS, AP, B, LDB, INFO)
ENTRY      spptrs (UPLO, N, NRHS, AP, B, LDB, INFO)
INTEGER    N, NRHS, LDB, INFO
real      AP(*), B(LDB,*)
CHARACTER*1 UPLO
```

The ENTRY statement enables the routine to be called by its LAPACK name.

3 Description

To solve a real symmetric positive-definite system of linear equations $AX = B$, this routine must be preceded by a call to F07GDF (SPPTRF/DPPTRF) which computes the Cholesky factorization of A using packed storage. The solution X is computed by forward and backward substitution.

If UPLO = 'U', $A = U^T U$, where U is upper triangular; the solution X is computed by solving $U^T Y = B$ and then $UX = Y$.

If UPLO = 'L', $A = LL^T$, where L is lower triangular; the solution X is computed by solving $LY = B$ and then $L^T X = Y$.

4 References

Golub G H and van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

1: UPLO – CHARACTER*1 *Input*

On entry: indicates whether A has been factorized as $U^T U$ or LL^T as follows:

if UPLO = 'U', $A = U^T U$, where U is upper triangular;

if UPLO = 'L', $A = LL^T$, where L is lower triangular.

Constraint: UPLO = 'U' or 'L'.

2: N – INTEGER *Input*

On entry: n , the order of the matrix A .

Constraint: $N \geq 0$.

- 3: NRHS – INTEGER *Input*
On entry: r , the number of right-hand sides.
Constraint: $\text{NRHS} \geq 0$.
- 4: AP(*) – *real* array *Input*
Note: the dimension of the array AP must be at least $\max(1, N * (N + 1)/2)$.
On entry: the Cholesky factor of A stored in packed form, as returned by F07GDF (SPPTRF/DPPTRF).
- 5: B(LDB,*) – *real* array *Input/Output*
Note: the second dimension of the array B must be at least $\max(1, \text{NRHS})$.
On entry: the n by r right-hand side matrix B .
On exit: the n by r solution matrix X .
- 6: LDB – INTEGER *Input*
On entry: the first dimension of the array B as declared in the (sub)program from which F07GEF (SPPTRS/DPPTRS) is called.
Constraint: $\text{LDB} \geq \max(1, N)$.
- 7: INFO – INTEGER *Output*
On exit: $\text{INFO} = 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

$\text{INFO} < 0$

If $\text{INFO} = -i$, the i th parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

For each right-hand side vector b , the computed solution x is the exact solution of a perturbed system of equations $(A + E)x = b$, where

if $\text{UPLO} = \text{'U'}$ then $|E| \leq c(n)\epsilon|U^T||U|$,

if $\text{UPLO} = \text{'L'}$ then $|E| \leq c(n)\epsilon|L||L^T|$,

$c(n)$ is a modest linear function of n , and ϵ is the *machine precision*.

If \hat{x} is the true solution, then the computed solution x satisfies a forward error bound of the form

$$\frac{\|x - \hat{x}\|_{\infty}}{\|x\|_{\infty}} \leq c(n) \text{cond}(A, x)\epsilon$$

where $\text{cond}(A, x) = \| |A^{-1}| |A| |x| \|_{\infty} / \|x\|_{\infty} \leq \text{cond}(A) = \| |A^{-1}| |A| \|_{\infty} \leq \kappa_{\infty}(A)$. Note that $\text{cond}(A, x)$ can be much smaller than $\text{cond}(A)$.

Forward and backward error bounds can be computed by calling F07GHF (SPPRFS/DPPRFS), and an estimate for $\kappa_{\infty}(A)$ ($= \kappa_1(A)$) can be obtained by calling F07GGF (SPPCON/DPPCON).

8 Further Comments

The total number of floating-point operations is approximately $2n^2r$.

This routine may be followed by a call to F07GHF (SPPRFS/DPPRFS) to refine the solution and return an error estimate.

The complex analogue of this routine is F07GSF (CPPTRS/ZPPTRS).

9 Example

To solve the system of equations $AX = B$, where

$$A = \begin{pmatrix} 4.16 & -3.12 & 0.56 & -0.10 \\ -3.12 & 5.03 & -0.83 & 1.18 \\ 0.56 & -0.83 & 0.76 & 0.34 \\ -0.10 & 1.18 & 0.34 & 1.18 \end{pmatrix} \quad \text{and} \quad B = \begin{pmatrix} 8.70 & 8.30 \\ -13.35 & 2.13 \\ 1.89 & 1.61 \\ -4.14 & 5.00 \end{pmatrix}.$$

Here A is symmetric positive-definite, stored in packed form, and must first be factorized by F07GDF (SPPTRF/DPPTF).

9.1 Program Text

Note: the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      F07GEF Example Program Text
*      Mark 15 Release. NAG Copyright 1991.
*      .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX, NRHMAX, LDB
      PARAMETER        (NMAX=8,NRHMAX=NMAX,LDB=NMAX)
*      .. Local Scalars ..
      INTEGER          I, IFAIL, INFO, J, N, NRHS
      CHARACTER        UPLO
*      .. Local Arrays ..
      real             AP(NMAX*(NMAX+1)/2), B(LDB,NRHMAX)
*      .. External Subroutines ..
      EXTERNAL         spptrf, spptrs, X04CAF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'F07GEF Example Program Results'
*      Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N, NRHS
      IF (N.LE.NMAX .AND. NRHS.LE.NRHMAX) THEN
*
*      Read A and B from data file
*
      READ (NIN,*) UPLO
      IF (UPLO.EQ.'U') THEN
        READ (NIN,*) ((AP(I+J*(J-1)/2),J=I,N),I=1,N)
      ELSE IF (UPLO.EQ.'L') THEN
        READ (NIN,*) ((AP(I+(2*N-J)*(J-1)/2),J=1,I),I=1,N)
      END IF
      READ (NIN,*) ((B(I,J),J=1,NRHS),I=1,N)
*
*      Factorize A
*
      CALL spptrf(UPLO,N,AP,INFO)
*
      WRITE (NOUT,*)
      IF (INFO.EQ.0) THEN
*
*      Compute solution
*
      CALL spptrs(UPLO,N,NRHS,AP,B,LDB,INFO)
```

```

*
*          Print solution
*
*          IFAIL = 0
*
*          CALL X04CAF('General',' ',N,NRHS,B,LDB,'Solution(s)',IFAIL)
*
*          ELSE
*            WRITE (NOUT,*) 'A is not positive-definite'
*          END IF
*        END IF
*      STOP
*
*    END

```

9.2 Program Data

F07GEF Example Program Data

```

  4  2          :Values of N and NRHS
  'L'          :Value of UPLO
  4.16
-3.12  5.03
  0.56 -0.83  0.76
-0.10  1.18  0.34  1.18  :End of matrix A
  8.70  8.30
-13.35  2.13
  1.89  1.61
-4.14  5.00          :End of matrix B

```

9.3 Program Results

F07GEF Example Program Results

```

Solution(s)
          1          2
1         1.0000      4.0000
2        -1.0000      3.0000
3         2.0000      2.0000
4        -3.0000      1.0000

```
